
Лабораторная работа 4 
По курсу: «Компьютерная схемотехника» 

 

Проектирование регистровой памяти 
 
 

1. Цель лабораторной работы:  
овладеть знаниями и практическими навыками по проектированию устройств памяти, 
таких как регистровое ОЗУ (FM). Лабораторная работа также служит для овладения 
навыками программирования и отладки описания RAM на языке VHDL. 

2. Теоретические сведения 
Блок FM (Fast Memory – быстрая память) предназначен для быстрого доступа и 

выдачи n-разрядных слов по нескольким произвольным адресам. В лабораторной работе 
предполагается, что данное имеет разрядность NR от 4 до 16 бит и объем памяти равен  М 
от 8 до 32 регистров. Каждый из каналов доступа к FM имеет свою адресную шину. 
Количество каналов доступа равно 2 или 3 и обозначаются буквами B,D и Q. Один из 
каналов предназначен для записи, а остальные – для чтения. Каналы могут быть  
двунаправленными, т.е. использоваться как для записи, так и для чтения. В лабораторной 
работе двунаправленный канал рекомендуется исполнить на основе шины с тремя 
состояниями, как в лабораторной работе 3. 

При проектировании микросхем, если объем RAM невелик, как в случае FM, то 
память набирают из отдельных триггеров.  

Запись данных в модуль FM всегда выполняется по фронту синхросерии или сигнала 
записи, т.е. вход модуля можно рассматривать как вход синхронного регистра.  Чтение 
данного выполняется сразу же после подачи адреса в FM.  

 
3. Элементная база 
 При выполнении лабораторной работы предлагается реализовать RAM на базе  

ПЛИС фирмы Xilinx. При этом используется такая же элементная база, как и в 
лабораторной работе 3. Там же описаны элементы, предназначенные для реализации 
тристабильных шин.  

 Кроме того, в библиотеку элементов ПЛИС фирмы Xilinx входят элементы типа 
RAM16X1D, представляющие собой FM объемом 16 однобитных слов с одним каналом для 
записи и чтения и другим каналом – для чтения. И поэтому эти элементы могут успешно 
использоваться при построении FM с двумя каналами. Этот элемент имеет следующий 
интерфейс. 

entity RAM16X1D is 
port (D : in std_ulogic; -- данное для записи по 1 каналу 
     WE : in std_ulogic; -- разрешение записи 
     WCLK: in std_ulogic;-- синхросигнал для записи 
     A0,A1,A2,A3: in std_ulogic;-- адрес 1 канала 
     DPRA0,DPRA1,DPRA2,DPRA3:in std_ulogic;-- адрес 2 канала 
     SPO : out std_ulogic;-- данное для чтения по 1 каналу 
     DPO : out std_ulogic -- данное для чтения по 2 каналу 
       );  
end RAM16X1D; 
 
4. Примеры описания FM 
Рассмотрим пример проектирования трехканальной FM объемом 8 

шестнадцатиразрядных слов.  Объявление объекта для такой FM выглядит следующим 
образом. 

use work.CNetlist.all; 
entity FM is 
 port(CLK:in BIT; -- синхровход 

      WR:in BIT;  -- сигнал записи 
      AB:in BIT_VECTOR(2 downto 0);-- адрес канала В  
      AD:in BIT_VECTOR(2 downto 0);-- адрес канала D   
      AQ:in BIT_VECTOR(2 downto 0);-- адрес канала Q 
      Q: in BIT_VECTOR (15 downto 0);-- данное канала Q 
      B: out BIT_VECTOR(15 downto 0);-- данное канала В 
      D: out BIT_VECTOR(15 downto 0));-- данное канала D 
end FM; 
4.1. Поведенческая модель FM 
Поведенческая модель FM во многом похожа на поведенческую модель RAM, 

описанную в лабораторной работе 3.  Отличия заключаются в том, что две параллельных 
операции чтения и запись выполняются по трем различным адресам, а регистр адреса и 
тристабильный буфер - отсутствуют.  
architecture BEH of FM is 

type MEM8X16 is array(0 to 7) of BIT_VECTOR(15 downto 0); 
signal addr,do: BIT_VECTOR(15 downto 0); 

begin 
FM8:process(CLK,AD,AB) ---- блок регистровой памяти --------------- 

       variable RAM: MEM8x16; 
       variable addrq,addrd,addrb:natural; 
    begin 
       addrq:= BIT_TO_INT(AQ); 
       addrd:= BIT_TO_INT(AD); 
       addrb:= BIT_TO_INT(AB); 
       if CLK='1' and CLK'event then 
              if WR = '1' then 
                 RAM(addrq):= Q;  -- запись 
              end if; 
       end if; 
       B<= RAM(addrb);         -- чтение канала В 
       D<= RAM(addrd);         -- чтение канала D 

    end process; 
end BEH;  

Данное описание относится к моделям, описанным синтезируемым стилем. И 
компилятор-синтезатор  выполняет эту память на отдельных триггерах.  

4.2. Структурная модель FM 
Рассмотрим проектирование FM на базе PLMT и трггеров. Блок FM должен 

содержать 8 регистров размером 16 бит, два мультиплексора считанных данных по 
каналам B,D и дешифратор записи по каналу Q (рис. 4.1).  

Рис.4.1. Структура блока FM 
 

Дешифратор DC состоит из восьми PLM, каждая из которых декодирует три разряда 
адреса. Первая из них описывается следующим образом при  D = WE, C =A2, B = A1, 
A=A0.  
architecture PLM_DC0 of PLM_4 is   

begin     
 Y<=(D and not C and not B and not A) after td;  

end PLM_DC0; 

Остальные PLM дешифратора - PLM_DC1,…,PLM_DC7- описываются аналогично. 
Каждый из мультиплексоров состоит из шестнадцати однобитных восьмивходовых 

мультиплексоров. Восьмивходовый мультиплексор можно представить как два 
четырехвходовых мультиплексора, выходы которых объединены в двухвходовом 
мультиплексоре. Четырехвходовый мультиплексор описан в лабораторной работе 3 как 
PLM с архитектурой of PLM_6(PLM_MUX). Двухвходовый мультиплексор имеет 
следующую архитектуру при  C = D1, B= D0, A=A0. 
architecture PLM_MUX of PLM_3 is   

begin     
 Y<=(B and not A)  -- 0-й вход  
     or (C and A)  -- 1-й вход 
       after td; --задержка элемента  
end PLM_MUX; 
Cтруктурная модель восьмивходового мультиплексора описывается в следующем 

объекте. 
entity MUX8 is port(D0,D1,D2,D3,D4,D5,D6,D7: in bit; -- входы данных 
                    A: in bit_vector(2 downto 0); -- адрес 
                    Q: out bit);                  -- выход данного 
end MUX8; 
architecture STR of MUX8 is 
 signal mux0,mux1:bit; 
 begin   
   U_MUX0: entity PLM_6(PLM_MUX)  
       port map(F=>D3,E=>D2,D=>D1,C=>D0,B=>A(1),A=>A(0),Y=>mux0); 
   U_MUX1: entity PLM_6(PLM_MUX)  
       port map(F=>D7,E=>D6,D=>D5,C=>D4,B=>A(1),A=>A(0),Y=>mux1); 
   U_MUX3: entity PLM_3(PLM_MUX)  
       port map(C=>mux1,B=>mux0,A=>A(2),Y=>Q); 
 end STR; 

 
Тело архитектуры регистровой памяти выглядит так. 

architecture STR of FM is 
   type FMARR is array(7 downto 0, 15 downto 0) of bit;  
   signal y:FMARR; 
   signal ce:bit_vector(7 downto 0); 
   constant gnd:bit:='0'; 
   component MUX8 is port(D0,D1,D2,D3,D4,D5,D6,D7: in bit; -- входы  
                    A: in bit_vector(2 downto 0); -- адрес 
                    Q: out bit);                  -- выход данного 
   end component; 
   component FDRE  is port (Q:out  bit; --триггер  

      D : in   bit; 
      C : in   bit; 
      CE: in   bit; 
      R : in   bit); 

   end component; 
begin   
   -- дешифратор адреса  (компоненты U_DC2,…, U_DC6 опущены)   
   U_DC0: entity PLM_4(PLM_DC0) 
          port map(D=>WR,C=>AQ(2),B=>AQ(1),A=>AQ(0),Y=>ce(0)); 
       ... 
   U_DC7: entity PLM_4(PLM_DC7) 
          port map(D=>WR,C=>AQ(2),B=>AQ(1),A=>AQ(0),Y=>ce(7)); 
   -- массив регистров   ------------------------ 
   U_FM: for i in 0 to 7 generate 
     U_RG: for j in 0 to 15 generate 
        U_TT: FDRE port map (D=>Q(j),  -- входное данное 

                        C =>CLK,   -- синхросигнал 
                        CE=> ce(i),  -- разрешение записи 
                        R => gnd,   -- сброс не используется 
                        Q=>y(i,j)); -- выходы триггеров  

       end generate; 
   end generate; 
-- выходные мультиплексоры  -------------------- 
U_MUX: for i in 0 to 15 generate 
     MUXD:MUX8 port map(D0=>y(0,i),D1=>y(1,i),D2=>y(2,i),D3=>y(3,i), 
                   D4=>y(4,i),D5=>y(5,i),D6=>y(6,i),D7=>y(7,i), 
                   A=>AD,        -- адрес 
                   Q=>D(i));     -- выход канала D 
     MUXB:MUX8 port map(D0=>y(0,i),D1=>y(1,i),D2=>y(2,i),D3=>y(3,i), 
                   D4=>y(4,i),D5=>y(5,i),D6=>y(6,i),D7=>y(7,i), 
                   A=>AB,        -- адрес 
                   Q=>B(i));     -- выход канала B 
   end generate; 
end STR; 

Для связи выходов триггеров FM с входами мультиплексоров используется сигнал у 
типа  двумерный массив размерами 8х16. 

4.3. Испытательный стенд для FM 
 Рассмотрим испытательный стенд - объект FM_TB - для архитектуры FM(STR), у 

которой эталонной моделью является архитектура RAM(BEH). Испытания заключаются в 
подаче случайных адресов, данных на вход DQ и периодического сигнала записи на обе 
модели и сравнении состояния  выходных шин данных моделей. Испытательный стенд 
аналогичен стенду в лабораторной работе 3. Но количество генераторов случайных чисел 
увеличено до  4 , а число компараторов – до 2.  

5. Порядок проведения лабораторной работы  
В соответствии с номером варианта, выбирается задание на выполнение 

лабораторной работы. Параметры задания включают: 
- тип логического элемента (PLM или LUT); 
- максимальное число термов PLM или количества входов LUT (4 или 5); 
- разрядность и объем результирующей схемы FM; 
- количество rканалов FM и их направленность. 
Выполнение лабораторной работы имеет 3 стадии: разработка поведенческой модели 

RAM, разработка структурной модели RAM и разработка испытательного стенда с 
проверкой функционирования RAM. 

6. Отчет по лабораторной работе. 
Отчет по лабораторной работе должен содержать: 
- цель работы,  
- описание варианта FM,  
- ход синтеза моделей PLM или содержимого LUT, 
- тексты описаний поведенческой и структурной моделей FM, 
- графики сигналов, снятых на испытательном стенде, 
- измеренные задержки сигналов, 
- выводы. 
 
7. Вопросы по лабораторной работе. 
Каково функциональное назначение FM? 
Какие элементы используют при проектировании FM? 
Какими способами задают FM в проектах на VHDL? 
От чего и насколько зависят аппаратные затраты FM? 
Почему в структурном проекте FM вставка компонента дешифратора выполнена как 

вставка объекта, а вставка триггера – как вставка компонента? 

